// File: contracts/erc20/ERC20.sol



// SPDX-License-Identifier: AGPL-3.0-only
pragma solidity >=0.8.0;

/// @notice Modern and gas efficient ERC20 + EIP-2612 implementation.
/// @author Solmate (https://github.com/transmissions11/solmate/blob/main/src/tokens/ERC20.sol)
/// @author Modified from Uniswap (https://github.com/Uniswap/uniswap-v2-core/blob/master/contracts/UniswapV2ERC20.sol)
/// @dev Do not manually set balances without updating totalSupply, as the sum of all user balances must not exceed it.
abstract contract ERC20 {
    /*//////////////////////////////////////////////////////////////
                                 EVENTS
    //////////////////////////////////////////////////////////////*/

    event Transfer(address indexed from, address indexed to, uint256 amount);

    event Approval(address indexed owner, address indexed spender, uint256 amount);

    /*//////////////////////////////////////////////////////////////
                            METADATA STORAGE
    //////////////////////////////////////////////////////////////*/

    string public name;

    string public symbol;

    uint8 public immutable decimals;

    /*//////////////////////////////////////////////////////////////
                              ERC20 STORAGE
    //////////////////////////////////////////////////////////////*/

    uint256 public totalSupply;

    mapping(address => uint256) public balanceOf;

    mapping(address => mapping(address => uint256)) public allowance;

    /*//////////////////////////////////////////////////////////////
                            EIP-2612 STORAGE
    //////////////////////////////////////////////////////////////*/

    uint256 internal immutable INITIAL_CHAIN_ID;

    bytes32 internal immutable INITIAL_DOMAIN_SEPARATOR;

    mapping(address => uint256) public nonces;

    /*//////////////////////////////////////////////////////////////
                               CONSTRUCTOR
    //////////////////////////////////////////////////////////////*/

    constructor(string memory _name, string memory _symbol, uint8 _decimals) {
        name = _name;
        symbol = _symbol;
        decimals = _decimals;

        INITIAL_CHAIN_ID = block.chainid;
        INITIAL_DOMAIN_SEPARATOR = computeDomainSeparator();
    }

    /*//////////////////////////////////////////////////////////////
                               ERC20 LOGIC
    //////////////////////////////////////////////////////////////*/

    function approve(address spender, uint256 amount) public virtual returns (bool) {
        allowance[msg.sender][spender] = amount;

        emit Approval(msg.sender, spender, amount);

        return true;
    }

    function transfer(address to, uint256 amount) public virtual returns (bool) {
        balanceOf[msg.sender] -= amount;

        // Cannot overflow because the sum of all user
        // balances can't exceed the max uint256 value.
        unchecked {
            balanceOf[to] += amount;
        }

        emit Transfer(msg.sender, to, amount);

        return true;
    }

    function transferFrom(address from, address to, uint256 amount) public virtual returns (bool) {
        uint256 allowed = allowance[from][msg.sender]; // Saves gas for limited approvals.

        if (allowed != type(uint256).max) allowance[from][msg.sender] = allowed - amount;

        balanceOf[from] -= amount;

        // Cannot overflow because the sum of all user
        // balances can't exceed the max uint256 value.
        unchecked {
            balanceOf[to] += amount;
        }

        emit Transfer(from, to, amount);

        return true;
    }

    /*//////////////////////////////////////////////////////////////
                             EIP-2612 LOGIC
    //////////////////////////////////////////////////////////////*/

    function permit(
        address owner,
        address spender,
        uint256 value,
        uint256 deadline,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) public virtual {
        require(deadline >= block.timestamp, "PERMIT_DEADLINE_EXPIRED");

        // Unchecked because the only math done is incrementing
        // the owner's nonce which cannot realistically overflow.
        unchecked {
            address recoveredAddress = ecrecover(
                keccak256(
                    abi.encodePacked(
                        "\x19\x01",
                        DOMAIN_SEPARATOR(),
                        keccak256(
                            abi.encode(
                                keccak256(
                                    "Permit(address owner,address spender,uint256 value,uint256 nonce,uint256 deadline)"
                                ),
                                owner,
                                spender,
                                value,
                                nonces[owner]++,
                                deadline
                            )
                        )
                    )
                ),
                v,
                r,
                s
            );

            require(recoveredAddress != address(0) && recoveredAddress == owner, "INVALID_SIGNER");

            allowance[recoveredAddress][spender] = value;
        }

        emit Approval(owner, spender, value);
    }

    function DOMAIN_SEPARATOR() public view virtual returns (bytes32) {
        return
            block.chainid == INITIAL_CHAIN_ID ? INITIAL_DOMAIN_SEPARATOR : computeDomainSeparator();
    }

    function computeDomainSeparator() internal view virtual returns (bytes32) {
        return
            keccak256(
                abi.encode(
                    keccak256(
                        "EIP712Domain(string name,string version,uint256 chainId,address verifyingContract)"
                    ),
                    keccak256(bytes(name)),
                    keccak256("1"),
                    block.chainid,
                    address(this)
                )
            );
    }

    /*//////////////////////////////////////////////////////////////
                        INTERNAL MINT/BURN LOGIC
    //////////////////////////////////////////////////////////////*/

    function _mint(address to, uint256 amount) internal virtual {
        totalSupply += amount;

        // Cannot overflow because the sum of all user
        // balances can't exceed the max uint256 value.
        unchecked {
            balanceOf[to] += amount;
        }

        emit Transfer(address(0), to, amount);
    }

    function _burn(address from, uint256 amount) internal virtual {
        balanceOf[from] -= amount;

        // Cannot underflow because a user's balance
        // will never be larger than the total supply.
        unchecked {
            totalSupply -= amount;
        }

        emit Transfer(from, address(0), amount);
    }
}

// File: contracts/erc20/SafeERC20.sol



// SPDX-License-Identifier: AGPL-3.0-only
pragma solidity >=0.8.0;

import "./ERC20.sol";

/// @notice Safe ETH and ERC20 transfer library that gracefully handles missing return values.
/// @author Solmate (https://github.com/transmissions11/solmate/blob/main/src/utils/SafeTransferLib.sol)
/// @dev Use with caution! Some functions in this library knowingly create dirty bits at the destination of the free memory pointer.
/// @dev Note that none of the functions in this library check that a token has code at all! That responsibility is delegated to the caller.
library SafeERC20 {
    /*//////////////////////////////////////////////////////////////
                             ETH OPERATIONS
    //////////////////////////////////////////////////////////////*/

    function safeTransferETH(address to, uint256 amount) internal {
        bool success;

        /// @solidity memory-safe-assembly
        assembly {
            // Transfer the ETH and store if it succeeded or not.
            success := call(gas(), to, amount, 0, 0, 0, 0)
        }

        require(success, "ETH_TRANSFER_FAILED");
    }

    /*//////////////////////////////////////////////////////////////
                            ERC20 OPERATIONS
    //////////////////////////////////////////////////////////////*/

    function safeTransferFrom(ERC20 token, address from, address to, uint256 amount) internal {
        bool success;

        /// @solidity memory-safe-assembly
        assembly {
            // Get a pointer to some free memory.
            let freeMemoryPointer := mload(0x40)

            // Write the abi-encoded calldata into memory, beginning with the function selector.
            mstore(
                freeMemoryPointer,
                0x23b872dd00000000000000000000000000000000000000000000000000000000
            )
            mstore(add(freeMemoryPointer, 4), from) // Append the "from" argument.
            mstore(add(freeMemoryPointer, 36), to) // Append the "to" argument.
            mstore(add(freeMemoryPointer, 68), amount) // Append the "amount" argument.

            success := and(
                // Set success to whether the call reverted, if not we check it either
                // returned exactly 1 (can't just be non-zero data), or had no return data.
                or(and(eq(mload(0), 1), gt(returndatasize(), 31)), iszero(returndatasize())),
                // We use 100 because the length of our calldata totals up like so: 4 + 32 * 3.
                // We use 0 and 32 to copy up to 32 bytes of return data into the scratch space.
                // Counterintuitively, this call must be positioned second to the or() call in the
                // surrounding and() call or else returndatasize() will be zero during the computation.
                call(gas(), token, 0, freeMemoryPointer, 100, 0, 32)
            )
        }

        require(success, "TRANSFER_FROM_FAILED");
    }

    function safeTransfer(ERC20 token, address to, uint256 amount) internal {
        bool success;

        /// @solidity memory-safe-assembly
        assembly {
            // Get a pointer to some free memory.
            let freeMemoryPointer := mload(0x40)

            // Write the abi-encoded calldata into memory, beginning with the function selector.
            mstore(
                freeMemoryPointer,
                0xa9059cbb00000000000000000000000000000000000000000000000000000000
            )
            mstore(add(freeMemoryPointer, 4), to) // Append the "to" argument.
            mstore(add(freeMemoryPointer, 36), amount) // Append the "amount" argument.

            success := and(
                // Set success to whether the call reverted, if not we check it either
                // returned exactly 1 (can't just be non-zero data), or had no return data.
                or(and(eq(mload(0), 1), gt(returndatasize(), 31)), iszero(returndatasize())),
                // We use 68 because the length of our calldata totals up like so: 4 + 32 * 2.
                // We use 0 and 32 to copy up to 32 bytes of return data into the scratch space.
                // Counterintuitively, this call must be positioned second to the or() call in the
                // surrounding and() call or else returndatasize() will be zero during the computation.
                call(gas(), token, 0, freeMemoryPointer, 68, 0, 32)
            )
        }

        require(success, "TRANSFER_FAILED");
    }

    function safeApprove(ERC20 token, address to, uint256 amount) internal {
        bool success;

        /// @solidity memory-safe-assembly
        assembly {
            // Get a pointer to some free memory.
            let freeMemoryPointer := mload(0x40)

            // Write the abi-encoded calldata into memory, beginning with the function selector.
            mstore(
                freeMemoryPointer,
                0x095ea7b300000000000000000000000000000000000000000000000000000000
            )
            mstore(add(freeMemoryPointer, 4), to) // Append the "to" argument.
            mstore(add(freeMemoryPointer, 36), amount) // Append the "amount" argument.

            success := and(
                // Set success to whether the call reverted, if not we check it either
                // returned exactly 1 (can't just be non-zero data), or had no return data.
                or(and(eq(mload(0), 1), gt(returndatasize(), 31)), iszero(returndatasize())),
                // We use 68 because the length of our calldata totals up like so: 4 + 32 * 2.
                // We use 0 and 32 to copy up to 32 bytes of return data into the scratch space.
                // Counterintuitively, this call must be positioned second to the or() call in the
                // surrounding and() call or else returndatasize() will be zero during the computation.
                call(gas(), token, 0, freeMemoryPointer, 68, 0, 32)
            )
        }

        require(success, "APPROVE_FAILED");
    }
}

// File: contracts/PerpetualBond/interfaces/IPerpetualBondDeployer.sol



// SPDX-License-Identifier: BUSL-1.1
pragma solidity ^0.8.0;

/**
 * @title Perpetual bond deployer interface
 * @author Bonding Finance
 */
interface IPerpetualBondDeployer {
    struct Parameters {
        address factory;
        address token;
    }

    function parameters() external view returns (address, address);
}

// File: contracts/PerpetualBond/interfaces/IPerpetualBondFactory.sol



// SPDX-License-Identifier: BUSL-1.1
pragma solidity ^0.8.0;

/**
 * @title Perpetual bond factory interface
 * @author Bonding Finance
 */
interface IPerpetualBondFactory {
    event VaultCreated(address indexed token, address vault);

    struct FeeInfo {
        address feeTo;
        uint256 vaultFee;
        uint256 surplusFee;
    }

    function paused() external view returns (bool);

    function feeInfo() external view returns (address feeTo, uint256 vaultFee, uint256 surplusFee);

    function allVaults(uint256) external view returns (address);

    function getVault(address token) external view returns (address);

    function allVaultsLength() external view returns (uint256);

    //////////////////////////
    /* Restricted Functions */
    //////////////////////////

    function createVault(address token) external returns (address vault);

    function setPaused(bool paused) external;

    function setStaking(address vault, address staking) external;

    function collectVaultFees(address vault) external;

    function collectSurplusFees(address staking) external;

    function setFeeTo(address feeTo) external;

    function setVaultFee(uint256 vaultFee) external;

    function setSurplusFee(uint256 surplusFee) external;
}

// File: contracts/PerpetualBond/interfaces/IPerpetualBondStaking.sol



// SPDX-License-Identifier: BUSL-1.1
pragma solidity ^0.8.0;

/**
 * @title Perpetual bond staking interface
 * @author Bonding Finance
 */
interface IPerpetualBondStaking {
    event Stake(address indexed user, address indexed token, uint256 amount);
    event Unstake(address indexed user, address indexed token, uint256 amount);
    event EmergencyWithdraw(address indexed user, address indexed token, uint256 amount);
    event Claim(address indexed user, address indexed token, uint256 amount);
    event Distribute(address indexed pool, uint256 rewardAmount, uint256 amountStaked);
    event CollectSurplus(address indexed feeTo, uint256 amount);

    struct PoolInfo {
        uint256 accRewardsPerShare;
        uint256 accRewards;
        uint256 claimedRewards;
    }

    struct UserInfo {
        uint256 amount;
        uint256 rewardDebt;
    }

    function factory() external view returns (address);

    function vault() external view returns (address);

    function yToken() external view returns (address);

    function lpToken() external view returns (address);

    function rewardToken() external view returns (address);

    function fees() external view returns (uint256);

    function userInfo(
        address token,
        address user
    ) external view returns (uint256 amount, uint256 rewardDebt);

    function poolInfo(
        address token
    )
        external
        view
        returns (uint256 accRewardsPerShare, uint256 accRewards, uint256 claimedRewards);

    function stake(address token, uint256 amount) external returns (uint256 rewards);

    function unstake(address token, uint256 amount) external returns (uint256 rewards);

    function emergencyWithdraw(address token) external;

    function harvest() external;

    function pendingRewards(address token, address user) external view returns (uint256 rewards);

    //////////////////////////
    /* Restricted Functions */
    //////////////////////////

    function collectFees(address feeTo) external;
}

// File: contracts/PerpetualBond/interfaces/IPerpetualBondToken.sol



// SPDX-License-Identifier: BUSL-1.1
pragma solidity ^0.8.0;

/**
 * @title Perpetual bond token interface
 * @author Bonding Finance
 */
interface IPerpetualBondToken {
    function vault() external view returns (address);

    //////////////////////////
    /* Restricted Functions */
    //////////////////////////

    function mint(address user, uint256 amount) external;

    function burn(address user, uint256 amount) external;
}

// File: contracts/PerpetualBond/interfaces/IPerpetualBondVault.sol



// SPDX-License-Identifier: BUSL-1.1
pragma solidity ^0.8.0;

/**
 * @title Perpetual bond vault interface
 * @author Bonding Finance
 */
interface IPerpetualBondVault {
    event Mint(address indexed user, uint256 amount);
    event Redeem(address indexed user, uint256 amount);
    event Harvest(uint256 amount);
    event CollectFees(address indexed feeTo, uint256 amount);

    function factory() external view returns (address);

    function token() external view returns (address);

    function dToken() external view returns (address);

    function yToken() external view returns (address);

    function staking() external view returns (address);

    function totalDeposits() external view returns (uint256);

    function fees() external view returns (uint256);

    function deposit(uint256 amount) external returns (uint256 mintAmount);

    function redeem(uint256 amount) external returns (uint256 redeemAmount);

    function harvest() external;

    function pendingRewards() external view returns (uint256);

    //////////////////////////
    /* Restricted Functions */
    //////////////////////////

    function setStaking(address staking) external;

    function collectFees(address feeTo) external;
}

// File: contracts/PerpetualBond/PerpetualBondDeployer.sol



// SPDX-License-Identifier: BUSL-1.1

pragma solidity ^0.8.0;

import "./interfaces/IPerpetualBondDeployer.sol";
import "./PerpetualBondVault.sol";

/**
 * @title Bonding Finance bond deployer contract
 * @author Bonding Finance
 */
contract PerpetualBondDeployer is IPerpetualBondDeployer {
    Parameters public override parameters;

    /**
     * @notice Deploys vault contract
     * @param factory Factory address
     * @param token Underlying token of the vault
     */
    function deploy(address factory, address token) internal returns (address vault) {
        parameters = Parameters({factory: factory, token: token});
        vault = address(new PerpetualBondVault{salt: keccak256(abi.encode(token))}());
        delete parameters;
    }
}

// File: contracts/PerpetualBond/PerpetualBondToken.sol



// SPDX-License-Identifier: BUSL-1.1
pragma solidity >=0.8.0;

import "./interfaces/IPerpetualBondToken.sol";
import "./PerpetualBondTokenImpl.sol";

/**
 * @title Perpetual bond token contract
 * @author Bonding Finance
 */
contract PerpetualBondToken is IPerpetualBondToken, PerpetualBondTokenImpl {
    address public immutable override vault;

    constructor(string memory _name, string memory _symbol, uint8 _decimals) {
        vault = msg.sender;
        name = _name;
        symbol = _symbol;
        decimals = _decimals;
    }

    //////////////////////////
    /* Restricted Functions */
    //////////////////////////

    function mint(address user, uint256 amount) external override {
        _onlyVault();

        _mint(user, amount);
    }

    function burn(address user, uint256 amount) external override {
        _onlyVault();

        _burn(user, amount);
    }

    function _onlyVault() internal view {
        require(msg.sender == vault, "!vault");
    }
}

// File: contracts/PerpetualBond/PerpetualBondTokenImpl.sol



// SPDX-License-Identifier: BUSL-1.1
pragma solidity >=0.8.0;

/**
 * @title Perpetual bond token abstract contract
 * @author Bonding Finance
 * @dev Modified from solmate (https://github.com/transmissions11/solmate/blob/main/src/tokens/ERC20.sol)
 */
abstract contract PerpetualBondTokenImpl {
    /*//////////////////////////////////////////////////////////////
                                 EVENTS
    //////////////////////////////////////////////////////////////*/

    event Transfer(address indexed from, address indexed to, uint256 amount);

    event Approval(address indexed owner, address indexed spender, uint256 amount);

    /*//////////////////////////////////////////////////////////////
                            METADATA STORAGE
    //////////////////////////////////////////////////////////////*/

    string public name;

    string public symbol;

    uint8 public decimals;

    /*//////////////////////////////////////////////////////////////
                              ERC20 STORAGE
    //////////////////////////////////////////////////////////////*/

    uint256 public totalSupply;

    mapping(address => uint256) public balanceOf;

    mapping(address => mapping(address => uint256)) public allowance;

    /*//////////////////////////////////////////////////////////////
                            EIP-2612 STORAGE
    //////////////////////////////////////////////////////////////*/

    uint256 internal immutable INITIAL_CHAIN_ID;

    bytes32 internal immutable INITIAL_DOMAIN_SEPARATOR;

    mapping(address => uint256) public nonces;

    /*//////////////////////////////////////////////////////////////
                               CONSTRUCTOR
    //////////////////////////////////////////////////////////////*/

    constructor() {
        INITIAL_CHAIN_ID = block.chainid;
        INITIAL_DOMAIN_SEPARATOR = computeDomainSeparator();
    }

    /*//////////////////////////////////////////////////////////////
                               ERC20 LOGIC
    //////////////////////////////////////////////////////////////*/

    function approve(address spender, uint256 amount) public virtual returns (bool) {
        allowance[msg.sender][spender] = amount;

        emit Approval(msg.sender, spender, amount);

        return true;
    }

    function transfer(address to, uint256 amount) public virtual returns (bool) {
        balanceOf[msg.sender] -= amount;

        // Cannot overflow because the sum of all user
        // balances can't exceed the max uint256 value.
        unchecked {
            balanceOf[to] += amount;
        }

        emit Transfer(msg.sender, to, amount);

        return true;
    }

    function transferFrom(address from, address to, uint256 amount) public virtual returns (bool) {
        uint256 allowed = allowance[from][msg.sender]; // Saves gas for limited approvals.

        if (allowed != type(uint256).max) allowance[from][msg.sender] = allowed - amount;

        balanceOf[from] -= amount;

        // Cannot overflow because the sum of all user
        // balances can't exceed the max uint256 value.
        unchecked {
            balanceOf[to] += amount;
        }

        emit Transfer(from, to, amount);

        return true;
    }

    /*//////////////////////////////////////////////////////////////
                             EIP-2612 LOGIC
    //////////////////////////////////////////////////////////////*/

    function permit(
        address owner,
        address spender,
        uint256 value,
        uint256 deadline,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) public virtual {
        require(deadline >= block.timestamp, "PERMIT_DEADLINE_EXPIRED");

        // Unchecked because the only math done is incrementing
        // the owner's nonce which cannot realistically overflow.
        unchecked {
            address recoveredAddress = ecrecover(
                keccak256(
                    abi.encodePacked(
                        "\x19\x01",
                        DOMAIN_SEPARATOR(),
                        keccak256(
                            abi.encode(
                                keccak256(
                                    "Permit(address owner,address spender,uint256 value,uint256 nonce,uint256 deadline)"
                                ),
                                owner,
                                spender,
                                value,
                                nonces[owner]++,
                                deadline
                            )
                        )
                    )
                ),
                v,
                r,
                s
            );

            require(recoveredAddress != address(0) && recoveredAddress == owner, "INVALID_SIGNER");

            allowance[recoveredAddress][spender] = value;
        }

        emit Approval(owner, spender, value);
    }

    function DOMAIN_SEPARATOR() public view virtual returns (bytes32) {
        return
            block.chainid == INITIAL_CHAIN_ID ? INITIAL_DOMAIN_SEPARATOR : computeDomainSeparator();
    }

    function computeDomainSeparator() internal view virtual returns (bytes32) {
        return
            keccak256(
                abi.encode(
                    keccak256(
                        "EIP712Domain(string name,string version,uint256 chainId,address verifyingContract)"
                    ),
                    keccak256(bytes(name)),
                    keccak256("1"),
                    block.chainid,
                    address(this)
                )
            );
    }

    /*//////////////////////////////////////////////////////////////
                        INTERNAL MINT/BURN LOGIC
    //////////////////////////////////////////////////////////////*/

    function _mint(address to, uint256 amount) internal virtual {
        totalSupply += amount;

        // Cannot overflow because the sum of all user
        // balances can't exceed the max uint256 value.
        unchecked {
            balanceOf[to] += amount;
        }

        emit Transfer(address(0), to, amount);
    }

    function _burn(address from, uint256 amount) internal virtual {
        balanceOf[from] -= amount;

        // Cannot underflow because a user's balance
        // will never be larger than the total supply.
        unchecked {
            totalSupply -= amount;
        }

        emit Transfer(from, address(0), amount);
    }
}

// File: contracts/PerpetualBond/PerpetualBondVault.sol



// SPDX-License-Identifier: BUSL-1.1

pragma solidity ^0.8.0;

import "./interfaces/IPerpetualBondVault.sol";
import "./interfaces/IPerpetualBondFactory.sol";
import "./interfaces/IPerpetualBondStaking.sol";
import "./PerpetualBondToken.sol";
import "./PerpetualBondDeployer.sol";
import "../erc20/SafeERC20.sol";
import "../erc20/ERC20.sol";
import "../erc20/ERC20.sol";
import "../utils/ReentrancyGuard.sol";

/**
 * @title Perpetual bond vault contract
 * @author Bonding Finance
 */
contract PerpetualBondVault is IPerpetualBondVault, ReentrancyGuard {
    using SafeERC20 for ERC20;

    address public immutable override factory;
    address public immutable override token;
    address public immutable override dToken;
    address public immutable override yToken;
    address public override staking;
    uint256 public override totalDeposits;
    uint256 public override fees;

    constructor() {
        (factory, token) = PerpetualBondDeployer(msg.sender).parameters();

        string memory tokenSymbol = ERC20(token).symbol();
        uint8 decimals = ERC20(token).decimals();

        dToken = address(
            new PerpetualBondToken{salt: keccak256(abi.encode(token))}(
                string(abi.encodePacked(tokenSymbol, " Deposit Token")),
                string(abi.encodePacked(tokenSymbol, "-D")),
                decimals
            )
        );

        yToken = address(
            new PerpetualBondToken{salt: keccak256(abi.encode(token))}(
                string(abi.encodePacked(tokenSymbol, " Yield Token")),
                string(abi.encodePacked(tokenSymbol, "-Y")),
                decimals
            )
        );
    }

    /**
     * @notice Deposit `amount` underlying tokens to mint `amount` dTokens and `amount` yTokens
     * @param amount Amount of underlying tokens to deposit
     * @return mintAmount Amount of dTokens and yTokens minted
     */
    function deposit(uint256 amount) external override nonReentrant returns (uint256 mintAmount) {
        _isNotPaused();

        ERC20(token).safeTransferFrom(msg.sender, address(this), amount);

        uint256 feeAmount = _chargeFee(amount);
        mintAmount = amount - feeAmount;
        totalDeposits += mintAmount;

        PerpetualBondToken(dToken).mint(msg.sender, mintAmount);
        PerpetualBondToken(yToken).mint(msg.sender, mintAmount);

        emit Mint(msg.sender, mintAmount);

        return mintAmount;
    }

    /**
     * @notice Redeem `amount` of dTokens and yTokens for underlying tokens
     * @param amount Amount of dTokens and yTokens to burn
     * @return redeemAmount The amount of underlying tokens redeemed
     */
    function redeem(uint256 amount) external override nonReentrant returns (uint256 redeemAmount) {
        _isNotPaused();

        PerpetualBondToken(dToken).burn(msg.sender, amount);
        PerpetualBondToken(yToken).burn(msg.sender, amount);
        totalDeposits -= amount;

        uint256 feeAmount = _chargeFee(amount);
        redeemAmount = amount - feeAmount;

        uint256 balance = ERC20(token).balanceOf(address(this));
        // In case of rounding error
        if (redeemAmount > balance) redeemAmount = balance;

        ERC20(token).safeTransfer(msg.sender, redeemAmount);

        emit Redeem(msg.sender, redeemAmount);

        return redeemAmount;
    }

    /**
     * @notice Collects pending rebase rewards and sends to staking contract to distribute
     * @dev Will return if `staking` or pending rewards is 0
     */
    function harvest() public override {
        if (staking == address(0)) return;

        uint256 amount = pendingRewards();
        if (amount == 0) return;

        ERC20(token).safeTransfer(staking, amount);

        emit Harvest(amount);
    }

    /**
     * @notice Calculates rebase yield that hasn't been distributed to staking contract
     * @return rewards Amount of pending rewards
     */
    function pendingRewards() public view override returns (uint256 rewards) {
        uint256 balance = ERC20(token).balanceOf(address(this));
        if (totalDeposits + fees >= balance) return 0;

        rewards = balance - totalDeposits - fees;
    }

    function _isNotPaused() internal view {
        require(!IPerpetualBondFactory(factory).paused(), "paused");
    }

    /**
     * @notice Applies vault fee (if any) to `amount`
     * @param amount The original amount
     * @return feeAmount The fee amount charged
     */
    function _chargeFee(uint256 amount) internal returns (uint256 feeAmount) {
        (, uint256 fee, ) = IPerpetualBondFactory(factory).feeInfo();
        if (fee == 0) return 0;

        feeAmount = (amount * fee) / 10000;
        fees += feeAmount;

        return feeAmount;
    }

    //////////////////////////
    /* Restricted Functions */
    //////////////////////////

    function setStaking(address _staking) external override {
        require(msg.sender == factory, "!factory");

        staking = _staking;
    }

    function collectFees(address feeTo) external override {
        require(msg.sender == factory, "!factory");

        if (fees == 0) return;

        ERC20(token).safeTransfer(feeTo, fees);

        emit CollectFees(feeTo, fees);

        delete fees;
    }
}

// File: contracts/utils/ReentrancyGuard.sol



// SPDX-License-Identifier: AGPL-3.0-only
pragma solidity >=0.8.0;

/// @notice Gas optimized reentrancy protection for smart contracts.
/// @author Solmate (https://github.com/transmissions11/solmate/blob/main/src/utils/ReentrancyGuard.sol)
/// @author Modified from OpenZeppelin (https://github.com/OpenZeppelin/openzeppelin-contracts/blob/master/contracts/security/ReentrancyGuard.sol)
abstract contract ReentrancyGuard {
    uint256 private locked = 1;

    modifier nonReentrant() virtual {
        require(locked == 1, "REENTRANCY");

        locked = 2;

        _;

        locked = 1;
    }
}